Notacja Backhusa-Naura

Notacja Backhusa-Naura dopuszcza następujące znaki (symbole metajęzykowe): 
<,> - nawiasy ostre określają definiowane pojęcie, 
 ::= - oznacza „jest zdefiniowane jako”
  | - oznacza lub. 

<C++program> ::= <program_body>

<program_body> ::= [<wspace>] <main_function> [<wspace>]  |  [<wspace>] <diferent_definitions> [<wspace>] <program_body>  |  <program_body> [<wspace>] <diferent_definitions> [<wspace>]

<main_function> ::= [<main_function_start>] <kw_main> [<wspace>] <left_right_parens> [<wspace>] <left_brace> [<wspace>] <function_instructions_block> [<wspace>] <right_brace>

<main_function_start> ::= <kw_void> <wspace>

<diferent_definitions> ::= <definition>  |  <definition> [<wspace>] <diferent_definitions>

<definition> ::= <const_definition>  |  <var_definition>  |  <function_declaration  |  <function_definition>

<const_definition> ::= <kw_const> <wspace> <type> <wspace> <const_identifier> [<wspace>] <equals> [<wspace>] <ident_value> [<wspace>] <semicolon>

<type> ::= <kw_bool>  |  <kw_int>  |  <kw_float>  |  <kw_string>

<const_identifier> ::= <bool_const_identifier>  |  <num_const_identifier>  |  <string_const_identifier>

<bool_const_identifier> ::= <identifier>

<num_const_identifier> ::= <identifier>

<string_const_identifier> ::= <identifier>

<ident_value> ::= <bool_value>  |  <int_value>  |  <float_value>  |  <string_value>

<bool_value> ::= <kw_false> | <kw_true>

<var_definition> ::= <type> <wspace> <ident_list>

<ident_list> ::= <ident_item> [<wspace>] <semicolon>  |  <ident_item> [<wspace>] <comma> [<wspace>] <ident_list>

<ident_item> ::= <var_identifier>  |  <var_identifier> [<wspace>] <equals> [<wspace>] <ident_value>  |  <array_declarator>  |  <array_declarator> [<wspace>] <equals> [<wspace>] <left_brace> [wspace>] <value_list> [<wspace>] <right_brace>

<var_identifier> ::= <bool_var_identifier>  |  <num_var_identifier>  |  <string_var_identifier>

<bool_var_identifier> ::= <identifier>

<num_var_identifier> ::= <identifier>

<string_var_identifier> ::= <identifier>

<array_declarator> ::= <var_array_identifier> [<wspace>] <left_bracket> [<wspace>] <uint_value> [<wspace>] <right_bracket> 

<var_array_identifier> ::= <bool_avar_identifier>  |  <num_avar_identifier>  |  <string_avar_identifier>

<bool_avar_identifier> ::= <identifier>

<num_avar_identifier> ::= <identifier>

<string_avar_identifier> ::= <identifier>

<value_list> ::= <ident_value>  |  <ident_value> [<wspace>] <comma> [<wspace>] <value_list>

<function_declaration> ::= <function_header> <semicolon>

<function_header> ::= <function_type> <wspace> <fun_identifier> [<wspace>] <function_params> [<wspace>] 

<function_type> ::= <kw_void>  |  <type>

<fun_identifier> ::= <identifier>

<function_params> ::= <left_right_parens>  |  <left_paren> [<wspace>] <declaration_params> [<wspace>] <right_paren> 

<declaration_params> ::= <type> <wspace> <identifier>  |  <type> <wspace> <identifier> [<wspace>] <comma> [<wspace>] <declaration_params>

<function_definition> ::= <function_header> <left_brace> [<wspace>] <function_instructions_block> [<wspace>] <right_brace>

<function_instructions_block> ::= <const_definition>  |  <var_definition>  |  <statement>  |  <var_definition> [<wspace>] <function_instructions_block> |  <const_definition> [<wspace>] <function_instructions_block> |  <statement> [<wspace>] <function_instructions_block>

<statement> ::= <assignment> [<wspace>] <semicolon> |  <return_statement> [<wspace>] <semicolon> |  <function_invoking> [<wspace>] <semicolon> |  <input_statement> [<wspace>] <semicolon> |  <output_statement> [<wspace>] <semicolon> |  <kw_break> [<wspace>] <semicolon> |  <conditional_statement> |  <iterative_statement>

<assignment> ::= <num_variable> [<wspace>] <double_plus>  |  <double_plus> [<wspace>] <num_variable>  |  <num_variable> [<wspace>] <double_minus>  |  <double_minus> [<wspace>] <num_variable>  |  <bool_equals_assignment>  |  <num_equals_assignment>  |  <string_equals_assignment>

<num_variable> ::= <num_var_identifier>  |  <num_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <uint_value> [<wspace>] <right_bracket>

<bool_equals_assignment> ::= <bool_variable> [<wspace>] <equals> [<wspace>] <bool_expression>  |  <bool_variable> [<wspace>] <equals> [<wspace>] <bool_equals_assignment>

<bool_variable> ::= <bool_var_identifier>  |  <bool_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <uint_value> [<wspace>] <right_bracket>

<bool_expression> ::= <bool_factor>  |  <bool_factor> [<wspace>] <logical_operator> [<wspace>] <bool_expression>

<bool_factor> ::= <kw_false>  |  <kw_true>  |  <bool_const_identifier>  |  <bool_variable>  |  <bool_fun_invoking>  |  <exclamation> [<wspace>] <bool_factor>  |  <left_paren> [<wspace>] <bool_expression> [<wspace>] <right_paren>  |  <left_paren> [<wspace>] <bool_equals_assignment> [<wspace>] <right_paren>  |  <bool_factor> [<wspace>] <bool_relation_operator> [<wspace>] <bool_factor>  |  <num_expression> [<wspace>] <num_relation_operator> [<wspace>] <num_expression>

<logical_operator> ::= <double_and>  |  <double_bar>

<bool_relation_operator> ::= <double_equals>  |  <neg_equals>

<num_relation_operator> ::= <double_equals>  |  <neg_equals>  |  <less>  |  <greater>  |  <less_equals>  |  <greater_equals>

<num_equals_assignment> ::= <num_variable> [<wspace>] <num_equals_operator> [<wspace>] <num_expression>  |  <num_variable> [<wspace>] <num_equals_operator> [<wspace>] <num_equals_assignment>

<num_equals_operator> ::= <equals>  |  <plus_equals>  |  <minus_equals>  |  <star_equals>  |  <slash_equals>  |  <percent_equals>

<num_expression> ::= <num_multip>  |  <num_multip> [<wspace>] <add_operator> [<wspace>] <num_expression>

<num_multip> ::= <num_factor>  |  <num_factor> [<wspace>] <mult_operator> [<wspace>] <num_multip>

<num_factor> ::= <int_value>  |  <float_value>  |  <num_const_identifier>  |  <num_variable>  |  <num_variable> [<wspace>] <double_plus>  |  <num_variable> [<wspace>] <double_minus>  |  <double_plus> [<wspace>] <num_variable>  |  <double_minus> [<wspace>] <num_variable>  |  <num_fun_invoking>  |  <sign> [<wspace>] <num_factor>  |  <left_paren> [<wspace>] <num_expression> [<wspace>] <right_paren>  |  <left_paren> [<wspace>] <num_equals_assignment> [<wspace>] <right_paren>

<add_operator> ::= <plus>  |  <minus>

<mult_operator> ::= <star>  |  <slash>  |  <percent>

<sign> ::= <plus>  |  <minus>

<string_equals_assignment> ::= <string_variable> [<wspace>] <string_equals_operator> [<wspace>] <string_expression>  |  <string_variable> [<wspace>] <string_equals_operator> [<wspace>] <string_equals_assignment>

<string_variable> ::= <string_var_identifier>  |  <string_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <uint_value> [<wspace>] <right_bracket>

<string_equals_operator> ::= <equals>  |  <plus_equals>

<string_expression> ::= <string_factor>  |  <string_factor> [<wspace>] <plus> [<wspace>] <string_expression>

<string_factor> ::= <string_value>  |  <string_const_identifier>  |  <string_variable>  |  <string_fun_invoking>  |  <left_paren> [<wspace>] <string_expression> [<wspace>] <right_paren>  |  <left_paren> [<wspace>] <string_equals_assignment> [<wspace>] <right_paren>

<function_invoking> ::= <bool_fun_invoking>  |  <num_fun_invoking>  |  <string_fun_invoking>

<bool_fun_invoking> ::= <bool_fun_identifier> [<wspace>] <left_paren> [<wspace>] <right_paren>  |  <bool_fun_identifier> [<wspace>] <left_paren> [<wspace>] <fun_args_list> [<wspace>] <right_paren>

<bool_fun_identifier> ::= <fun_identifier>

<num_fun_invoking> ::= <num_fun_identifier> [<wspace>] <left_paren> [<wspace>] <right_paren>  |  <num_fun_identifier> [<wspace>] <left_paren> [<wspace>] <fun_args_list> [<wspace>] <right_paren>

<num_fun_identifier> ::= <fun_identifier>

<string_fun_invoking> ::= <string_fun_identifier> [<wspace>] <left_paren> [<wspace>] <right_paren>  |  <string_fun_identifier> [<wspace>] <left_paren> [<wspace>] <fun_args_list> [<wspace>] <right_paren>

<string_fun_identifier> ::= <fun_identifier>

<fun_args_list> ::= <expression>  |  <expression> [<wspace>] <comma> [<wspace>] <fun_args_list>

<expression> ::= <bool_expression>  |  <bool_equals_assignment>  |  <num_expression>  |  <num_equals_assignment>  |  <string_expression>  |  <string_equals_assignment>

<return_statement> ::= <kw_return>  |  <kw_return> <wspace> <expression>

<input_statement> ::= <kw_cin> [<wspace>] <double_greater> [<wspace>] <input_list>

<input_list> ::= <variable>  |  <variable> [<wspace>] <double_greater> [<wspace>] <input_list>

<variable> ::= <bool_variable>  |  <num_variable>  |  <string_variable>

<bool_variable> ::= <bool_var_identifier>  |  <bool_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <num_equals_assignment> [<wspace>] <right_bracket>  |  <bool_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <num_expression> [<wspace>] <right_bracket>

<num_variable> ::= <num_var_identifier>  |  <num_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <num_equals_assignment> [<wspace>] <right_bracket>  |  <num_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <num_expression> [<wspace>] <right_bracket>

<string_variable> ::= <string_var_identifier>  |  <string_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <num_equals_assignment> [<wspace>] <right_bracket>  |  <string_avar_identifier> [<wspace>] <left_bracket> [<wspace>] <num_expression> [<wspace>] <right_bracket>

<output_statement> ::= <kw_cout> [<wspace>] <double_less> [<wspace>] <output_list>

<output_list> ::= <output_item>  |  <output_item> [<wspace>] <double_less> [<wspace>] <output_list>

<output_item> ::= <kw_endl>  |  <expression>

<conditional_statement> ::= <if_statement>  |  <switch_statement>

<if_statement> ::= <if_then_statement>  |  <if_then_else_statement>

<if_then_statement> ::= <kw_if> [<wspace>] <left_paren> [<wspace>] <bool_expression> [<wspace>] <right_paren> [<wspace>] <instructions_block>

<instructions_block> ::= <statement>  |  <left_brace> [<wspace>] <instructions> [<wspace>] <right_brace>

<instructions> ::= <statement>  |  <statement> [<wspace>] <instructions>

<if_then_else_statement> ::= <if_then_statement> [<wspace>] <else_instructions_block>

<else_instructions_block> ::= <kw_else> <wspace> <statement>  |  <kw_else> [<wspace>] <left_brace> [<wspace>] <instructions> [<wspace>] <right_brace>

<switch_statement> ::= <kw_switch> [<wspace>] <left_paren> [<wspace>] <num_expression> [<wspace>] <right_paren> [<wspace>] <left_brace> [<wspace>] <cases> <right_brace>

<cases> ::= <cases_without_default>  |  <cases_with_default>

<cases_without_default> ::= <case_branch>  |  <case_branch> <cases_without_default>

<cases_with_default> ::= <cases_without_default> <kw_default> [<wspace>] <colon> [<wspace>] <instructions> [<wspace>]

<case_branch> ::= <kw_case> <wspace> <num_expression> [<wspace>] <colon> [<wspace>] [<instructions>] [<wspace>]

<iterative_statement> ::= <for_statement>  |  <while_statement>  |  <do_while_statement>

<for_statement> ::= <kw_for> [<wspace>] <left_paren> [<wspace>] [<for_var_definitions>] <semicolon> [<wspace>] [<bool_expression>] [<wspace>] <semicolon> [<wspace>] [<for_expression>] <right_paren> [<wspace>] <instructions_block>

<for_var_definitions> ::= <num_type> <wspace> <for_ident_list> [<wspace>]  |  <for_ident_list> [<wspace>]

<num_type> ::= <kw_int>  |  <kw_float>

<for_ident_list> ::= <num_identifier> [<wspace>] <equals> [<wspace>] <num_value>  |  <num_identifier> [<wspace>] <equals> [<wspace>] <num_value> [<wspace>] <coma> [<wspace>] <for_ident_list>

<num_value> ::= <int_value>  |  <float_value>

<for_expression> ::= <num_expression> [<wspace>]  |  <num_expression> [<wspace>] <coma> [<wspace>] <for_expression>

<while_statement> ::= <kw_while> [<wspace>] <left_paren> [<wspace>] <bool_expression> [<wspace>] <right_paren> [<wspace>] <instructions_block>

<do_while_statement> ::= <do_instructions_block> [<wspace>] <kw_while> [<wspace>] <left_paren> [<wspace>] <bool_expression> [<wspace>] <right_paren> [<wspace>] <semicolon>

<do_instructions_block> ::= <kw_do> <wspace> <statement>  |  <kw_do> [<wspace>] <left_brace> [<wspace>] <instructions> [<wspace>] <right_brace>

<wspace> ::= <space>  |  <tab>  |  <crlf>  |  <comment>  |  <space> <wspace>  |  <tab> <wspace>  |  <crlf> <wspace>  |  <comment> <wspace>

<comment> ::= "/*" <chars> "*/"  |  "//" <chars> <crlf>

<chars> ::= <char>  |  <char> <chars>

<space> ::= " "

<tab> ::= "
"

<crlf> := "\r\n"

<identifier> ::= <letter  |  <underline>  |  <identifier> <id_char>

<id_char> ::= <letter>  |  <digit>  |  <underline>

<int_value> ::= [<sign>] <uint_value>

<uint_value> ::= <digit>  |  <digit> <uint_value>

<float_value> ::= <int_value>  |  <int_value> <periot> <uint_value>

<string_value> ::= <quotation> [char_string>] <quotation>

<char_string> ::= <char>  |  <char> <char_string>

<kw_void> ::= <v><o><i><d>

<kw_main> ::= <m><a><i><n>

<kw_const> ::= <c><o><n><s><t>

<kw_bool> ::= <b><o><o><l>

<kw_false> ::= <f><a><l><s><e>

<kw_true> ::= <t><r><u><e>

<kw_int> ::= <i><n><t>

<kw_float> ::= <f><l><o><a><t>

<kw_string> ::= <s><t><r><i><n><g>

<kw_return> ::= <r><e><t><u><r><n>

<kw_cin> ::= <c><i><n>

<kw_cout> ::= <c><o><u><t>

<kw_endl> ::= <e><n><d><l>

<kw_if> ::= <i><f>

<kw_else> ::= <e><l><s><e>

<kw_switch> ::= <s><w><i><t><c><h>

<kw_case> ::= <c><a><s><e>

<kw_default> ::= <d><e><f><a><u><l><t>

<kw_break> ::= <b><r><e><a><k>

<kw_for> ::= <f><o><r>

<kw_while> ::= <w><h><i><l><e>

<kw_do> ::= <d><o>

<letter> ::= a|b|c|d|e|f|g|h|i|j|k|l|m|n|o|p|q|r|s|t|u|v|w|x|y|z|A|B|C|

D|E|F|G|H|I|J|K|L|M|N|O|P|Q|R|S|T|U|V|W|X|Y|Z

<digit> ::= 0|1|2|3|4|5|6|7|8|9

<underline> ::= "_"

<quotation> ::= """

<left_brace> ::= "{"

<right_brace> ::= "}"

<left_bracket> ::= "["

<right_bracket> ::= "]"

<left_paren> ::= "("

<right_paren> ::= ")"

<left_right_parens> ::= "()"

<colon> ::= ":"

<semicolon> ::= ";"

<comma> ::= ","

<periot> ::= "."

<equals> ::= "="

<less> ::= "<"

<greater> ::= ">"

<double_less> ::= "<<"

<double_greater> ::= ">>"

<star> ::= "*"

<slash> ::= "/"

<percent> ::= "%"

<plus> ::= "+"

<minus> ::= "-"

<double_plus> ::= "++"

<double_minus> ::= "--"

<neg_equals> ::= "!="

<less_equals> ::= "<="

<greater_equals> ::= ">="

<double_and> ::= "&&"

<double_bar> ::= "||"

<exclamation> ::= "!"

<plus_equals> ::= "+="

<minus_equals> ::= "-="

<star_equals> ::= "*="

<slash_equals> ::= "/="

<percent_equals> ::= "%="

<a> ::= "a" | "A"

<b> ::= "b" | "B"

<c> ::= "c" | "C"

<d> ::= "d" | "D"

<e> ::= "e" | "E"

<f> ::= "f" | "F"

<g> ::= "g" | "G"

<h> ::= "h" | "H"

<i> ::= "i" | "I"

<j> ::= "j" | "J"

<k> ::= "k" | "K"

<l> ::= "l" | "L"

<m> ::= "m" | "M"

<n> ::= "n" | "N"

<o> ::= "o" | "O"

<p> ::= "p" | "P"

<q> ::= "q" | "Q"

<r> ::= "r" | "R"

<s> ::= "s" | "S"

<t> ::= "t" | "T"

<u> ::= "u" | "U"

<v> ::= "v" | "V"

<w> ::= "w" | "W"

<x> ::= "x" | "X"

<y> ::= "y" | "Y"

<z> ::= "z" | "

